GAPを使ってみよう

星　明考

早稲田大学 教育学部

2008年1月16日

GAP とは？

GAP(Groups, Algorithms and Programming),
Version 4.4.10; 2007
free software
http://www.gap-system.org/ から無料で入手可能

```
gap>
```

が現われたら，入力待ち状態になる

```
gap> ?group
```

の様に？ キーワード でヘルプが見れる

```
gap> LogTo("log.txt");
```

としてファイル"log.txt"にログをとる事ができる

簡単な計算 (1)

簡単な計算で GAP に慣れる

```
gap> 1+1;
2
gap> 2^100;
1267650600228229401496703205376
gap> Factorial(30);
265252859812191058636308480000000
gap> 2^10 mod 11;
1
gap> 5=5;
true
gap> 5<4;
false
```

(Factorial=階乗) $n!$

簡単な計算 (2)

```
gap> Primes[1];
2
gap> List([1..30],x->Primes[x]);
[ 2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41,
  43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97,
  101, 103, 107, 109, 113 ]
gap> Factors(100);
[ 2, 2, 5, 5 ]
gap> Product([1..10]);
3628800
gap> Sum([1..10]);
55
gap> Filtered([1..100],x->IsPrime(x)=true);
[ 2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41,
  43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97 ]
```

置換と作用 (1)

巡回置換を定義する

```
gap> s:=(1,2,3,4,5);
(1,2,3,4,5)
gap> s^2;
(1,3,5,2,4)
gap> for i in [1..5] do;
> Print(s^i,"\n"); od;
(1,2,3,4,5)
(1,3,5,2,4)
(1,4,2,5,3)
(1,5,4,3,2)
()
gap> s^-1;
(1,5,4,3,2)
```

置換と作用 (2)

```
gap> s;
(1,2,3,4,5)
gap> t:=(1,2,3);
(1,2,3)
gap> s*t;
(1,3,4,5,2)
gap> t*s;
(1,3,2,4,5)
gap> 2^t;
3
gap> (2^s)^t;
1
```

GAP は積を左から計算するので注意が必要！(右作用)

群を定義する (1)

n 次対称群 S_n を定義する (対称群=Symmetric group)

```
gap> s3:=SymmetricGroup(3);
Sym( [ 1 .. 3 ] )
gap> Elements(s3);
[ (), (2,3), (1,2), (1,2,3), (1,3,2), (1,3) ]
gap> List(Elements(s3),x->Order(x));
[ 1, 2, 2, 3, 3, 2 ]
gap> Size(s3);
6
gap> GeneratorsOfGroup(s3);
[ (1,2,3), (1,2) ]
gap> t3:=Group((1,2,3),(1,2));
Group([ (1,2,3), (1,2) ])
gap> s3=t3;
true
```

群を定義する (2)

```
gap> s4:=SymmetricGroup(4);
Sym( [ 1 .. 4 ] )
gap> s5:=SymmetricGroup(5);;
gap> s6:=SymmetricGroup(6);;
gap> s7:=SymmetricGroup(7);;
```

交代群 A_n を定義 (交代群=Alternating group)

```
gap> a3:=AlternatingGroup(3);
Alt( [ 1 .. 3 ] )
gap> Elements(a3);
[ (), (1,2,3), (1,3,2) ]
gap> a4:=AlternatingGroup(4);;
gap> a5:=AlternatingGroup(5);;
gap> a6:=AlternatingGroup(6);;
gap> a7:=AlternatingGroup(7);;
```

群の計算 (1)

有限群を具体的に見てみる

```
gap> Elements(a4);
[ (), (2,3,4), (2,4,3), (1,2)(3,4), (1,2,3),
  (1,2,4), (1,3,2), (1,3,4), (1,3)(2,4),
  (1,4,2), (1,4,3), (1,4)(2,3) ]
gap> IsSubgroup(s5,s4);
true
gap> IsSubgroup(a5,s4);
false
gap> Size(a7);
2520
gap> Factorial(7)/2;
2520
gap> Intersection(a5,s4)=a4;
true
```

群の計算 (2)

二面体群 D_n を定義

```
gap> d4:=Group((1,2,3,4),(1,4)(2,3));;
gap> d5:=Group((1,2,3,4,5),(1,4)(2,3));;
gap> d6:=Group((1,2,3,4,5,6),(1,5)(2,4));;
gap> Elements(d5);
[ (), (2,5)(3,4), (1,2)(3,5), (1,2,3,4,5),
  (1,3)(4,5), (1,3,5,2,4), (1,4)(2,3),
  (1,4,2,5,3), (1,5,4,3,2), (1,5)(2,4) ]
gap> Size(d5);
10
gap> IsSubgroup(s5,d5);
true
gap> IsNormal(s5,d5);
false
```

D_5 は S_5 の部分群であるが，正規部分群ではない

群の計算 (3)

巡回群 C_n を定義

```
gap> c2:=Group((1,2));
Group([ (1,2) ])
gap> c3:=Group((1,2,3));;
gap> c4:=Group((1,2,3,4));;
gap> c5:=Group((1,2,3,4,5));;
gap> c6:=Group((1,2,3,4,5,6));;
gap> Elements(c4);
[ (), (1,2,3,4), (1,3)(2,4), (1,4,3,2) ]
gap> IsCyclic(c5);
true
gap> IsAbelian(c5);
true
gap> IsAbelian(d5);
false
```

群の計算 (4)

クラインの四元群 V_4 を定義

```
gap> v4:=Group((1,2)(3,4),(1,3)(2,4),(1,4)(2,3));
Group([ (1,2)(3,4), (1,3)(2,4), (1,4)(2,3) ])
gap> Size(v4);
4
gap> Elements(v4);
[ (), (1,2)(3,4), (1,3)(2,4), (1,4)(2,3) ]
gap> IsCyclic(v4);
false
gap> IsAbelian(v4);
true
gap> IsNormal(s4,v4);
true
```

$V_4 \triangleleft S_4$ かつ $S_4/V_4 \cong S_3$ であった

右剰余類

右剰余類の集合 $H \backslash G$ を求める (右剰余類=right coset)

```
gap> RightCosets(s3,c2);
[ RightCoset(Group( [ (1,2) ] ),()),
  RightCoset(Group( [ (1,2) ] ),(1,3)),
  RightCoset(Group( [ (1,2) ] ),(1,3,2)) ]
gap> List(RightCosets(s3,c2),x->Representative(x));
[ (), (1,3), (1,3,2) ]
gap> List(RightCosets(s4,v4),x->Representative(x));
[ (), (3,4), (2,3), (2,3,4), (2,4,3), (2,4) ]
gap> List(RightCosets(s4,c2),x->Representative(x));
[ (), (2,4), (1,2,4), (1,3), (1,3)(2,4), (1,2,4,3),
  (1,3,2), (1,3,2,4), (2,4,3), (1,3,4), (1,3,4,2),
  (1,2)(3,4) ]
```

$H \backslash G = \{Hg_1, \ldots, Hg_k\}$ の完全代表系 (代表元=representative)

剰余群 (商群)

GAP では剰余群 $H = G/N$ も手軽に扱える

```
gap> h:=s4/v4;
Group([ f1, f2 ])
gap> IsGroup(h);
true
gap> Elements(h);
[ <identity> of ..., f1, f2, f1*f2, f2^2, f1*f2^2 ]
gap> Size(h);
6
gap> IsomorphismGroups(h,c6);
fail
gap> IsomorphismGroups(h,s3);
[ f1, f2 ] -> [ (2,3), (1,2,3) ]
```

群の同型を判定し，$S_4/V_4 \cong S_3$ の同型写像を具体的に与える (同型=isomorphism)

部分群の計算 (1)

全ての部分群を得るには sonata パッケージを読み込む

```
gap> LoadPackage("sonata");
true
gap> Subgroups(s3);
[ Group(()), Group([ (2,3) ]), Group([ (1,2) ]),
  Group([ (1,3) ]), Group([ (1,2,3) ]),
  Group([ (1,3,2), (1,2) ]) ]
gap> List(last,x->Size(x));
[ 1, 2, 2, 2, 3, 6 ]
```



部分群の計算 (2)

```
gap> Subgroups(a4);
[ Group(()), Group([ (1,2)(3,4) ]),
  Group([ (1,3)(2,4) ]), Group([ (1,4)(2,3) ]),
  Group([ (2,4,3) ]), Group([ (1,3,4) ]),
  Group([ (1,4,2) ]), Group([ (1,2,3) ]),
  Group([ (1,3)(2,4), (1,2)(3,4) ]),
  Group([ (1,3)(2,4), (1,4)(2,3), (2,4,3) ]) ]
gap> List(last,x->Size(x));
[ 1, 2, 2, 2, 3, 3, 3, 3, 4, 12 ]
```

A_4 には位数 6 の部分群はない

部分群の計算 (3)

```
gap> Subgroups(d5);
[ Group(()), Group([ (2,5)(3,4) ]),
  Group([ (1,4)(2,3) ]), Group([ (1,2)(3,5) ]),
  Group([ (1,5)(2,4) ]), Group([ (1,3)(4,5) ]),
  Group([ (1,2,3,4,5) ]),
  Group([ (1,5,4,3,2), (1,4)(2,3) ]) ]
gap> List(last,x->Size(x));
[ 1, 2, 2, 2, 2, 2, 5, 10 ]
```



部分群の計算 (4)

D_4, S_4, A_5 の部分群の様子を調べてみる

```
gap> List(Subgroups(d4),x->Size(x));
[ 1, 2, 2, 2, 2, 2, 4, 4, 4, 8 ]
gap> List(Subgroups(s4),x->Size(x));
[ 1, 2, 2, 2, 2, 2, 2, 2, 2, 2, 3, 3, 3, 3, 4,
  4, 4, 4, 4, 4, 4, 6, 6, 6, 6, 8, 8, 8, 12, 24 ]
gap> List(Subgroups(a5),x->Size(x));
[ 1, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2,
  3, 3, 3, 3, 3, 3, 3, 3, 3, 3, 4, 4, 4, 4, 4, 5,
  5, 5, 5, 5, 5, 6, 6, 6, 6, 6, 6, 6, 6, 6, 6, 10,
  10, 10, 10, 10, 10, 12, 12, 12, 12, 12, 60 ]
```

A_5 には位数 $15, 20, 30$ の部分群は存在しない

共役類の計算 (1)

共役作用 $G \times G \to G, (g,x) \mapsto gxg^{-1}$ による軌道 $x^G := \mathrm{Orb}_G(x) = \{gxg^{-1} \mid g \in G\}$ を x の**共役類**, 共役類の類別 $G = \bigcup C_i$ に対し $|G| = |C_1| + \cdots + |C_k|$ を**類等式**といった

```
gap> ConjugacyClasses(s3);
[ ()^G, (1,2)^G, (1,2,3)^G ]
gap> ConjugacyClasses(s4);
[ ()^G, (1,2)^G, (1,2)(3,4)^G, (1,2,3)^G,
  (1,2,3,4)^G ]
gap> List(ConjugacyClasses(s4),x->Size(x));
[ 1, 6, 3, 8, 6 ]
gap> List(ConjugacyClasses(s5),x->Size(x));
[ 1, 10, 15, 20, 20, 30, 24 ]
```

S_3, S_4, S_5 の共役類 (クラス) の数はそれぞれ 3, 5, 7

共役類の計算 (2)

n	1	2	3	4	5	6	7	8	9	10
S_n の共役類の数	1	2	3	5	7	11	15	22	30	42

11	12	13	14	15	16	17	18	19	20	21
56	77	101	135	176	231	297	385	490	627	792

22	23	24	25	26	27	28	29	30
1002	1255	1575	1958	2436	3010	3718	4565	5604

命題. S_n の元 σ, τ が共役 $\iff$ σ と τ は共通の数字を含まない巡回置換の積で表したとき同じ型である

$$S_3 = \{(1)\} \cup \{(1\ 2), (1\ 3), (2\ 3)\} \cup \{(1\ 2\ 3), (1\ 3\ 2)\}$$

$$\begin{aligned} S_4 = &\{(1)\} \cup \{(3\ 4), (2\ 3), (2\ 4), (1\ 2), (1\ 3), (1\ 4)\} \\ &\cup \{(1\ 2)(3\ 4), (1\ 3)(2\ 4), (1\ 4)(2\ 3)\} \\ &\cup \{(2\ 3\ 4), (2\ 4\ 3), (1\ 2\ 3), (1\ 2\ 4), (1\ 3\ 2), (1\ 3\ 4), (1\ 4\ 2), (1\ 4\ 3)\} \\ &\cup \{(1\ 2\ 3\ 4), (1\ 2\ 4\ 3), (1\ 3\ 4\ 2), (1\ 3\ 2\ 4), (1\ 4\ 3\ 2), (1\ 4\ 2\ 3)\} \end{aligned}$$

分割数 (1)

S_5 の元	巡回置換の積の型	個数
(1)	(1,1,1,1,1)	1
(1 2)	(2,1,1,1)	${}_5C_2 = 10$
(1 2)(3 4)	(2,2,1)	${}_5C_4 \times 3 = 15$
(1 2 3)	(3,1,1)	${}_5C_3 \times 2 = 20$
(1 2 3)(4 5)	(3,2)	${}_5C_3 \times 2 = 20$
(1 2 3 4)	(4,1)	${}_5C_4 \times 3! = 30$
(1 2 3 4 5)	(5)	$4! = 24$

$$120 = 1 + 10 + 15 + 20 + 20 + 30 + 24 \quad (S_5\text{の類等式})$$

定義. $n = r_1 + \cdots + r_k, r_1 \geq \cdots \geq r_k$ なる組 $(r_1, \ldots, r_k)$ を**整数の分割** (integer partitions) という.

[参考文献] 整数の分割, G. アンドリュース, K. エリクソン
(佐藤文広 訳), 数学書房, 2006.

分割数 (2)

```
gap> Partitions(3);
[ [ 1, 1, 1 ], [ 2, 1 ], [ 3 ] ]
gap> Partitions(4);
[ [ 1, 1, 1, 1 ], [ 2, 1, 1 ], [ 2, 2 ], [ 3, 1 ],
  [ 4 ] ]
gap> Partitions(5);
[ [ 1, 1, 1, 1, 1 ], [ 2, 1, 1, 1 ], [ 2, 2, 1 ],
  [ 3, 1, 1 ], [ 3, 2 ],   [ 4, 1 ], [ 5 ] ]
gap> Length(Partitions(5));
7
gap> List([1..30],x->Length(Partitions(x)));
[ 1, 2, 3, 5, 7, 11, 15, 22, 30, 42, 56, 77, 101,
  135, 176, 231, 297, 385, 490, 627, 792, 1002,
  1255, 1575, 1958, 2436, 3010, 3718, 4565, 5604 ]
```

共役類の計算 (3)

$$1 = \frac{1}{120} + \frac{1}{12} + \frac{1}{8} + \frac{1}{6} + \frac{1}{6} + \frac{1}{4} + \frac{1}{5}$$
$$1 = \frac{1}{720} + \frac{1}{48} + \frac{1}{16} + \frac{1}{48} + \frac{1}{18} + \frac{1}{6} + \frac{1}{18} + \frac{1}{8} + \frac{1}{8} + \frac{1}{5} + \frac{1}{6}$$

```
gap> List(ConjugacyClasses(s5),x->Size(x))/Size(s5);
[ 1/120, 1/12, 1/8, 1/6, 1/6, 1/4, 1/5 ]
gap> List(ConjugacyClasses(s6),x->Size(x))/Size(s6);
[ 1/720, 1/48, 1/16, 1/48, 1/18, 1/6, 1/18,
  1/8, 1/8, 1/5, 1/6 ]
```

Fact. 固定した自然数 k に対し, $1 = \frac{1}{m_1} + \cdots + \frac{1}{m_k}$, $(m_1 \geq \cdots \geq m_k)$ の整数解は有限個であり, 共役類数 $h(G) = k$ の有限群 G も (同型を同一視して) 有限個しかない

定理 (演習課題) $h(G) = 3 \iff G \cong C_3, S_3$,
$h(G) = 4 \iff G \cong C_4, V_4, D_5, A_4$,
$h(G) = 5 \iff G \cong C_5, D_7, F_{21}, S_4, A_5, D_4, Q_8, F_{20}$

有限群の表現論, 既約表現の言葉で言うと？

共役類の計算 (4)

交代群 A_n の共役類分割

```
gap> cls4:=ConjugacyClasses(s4);;
gap> cls5:=ConjugacyClasses(s5);;
gap> List(cls4,x->Size(x));
[ 1, 6, 3, 8, 6 ]
gap> List(cls4,x->SignPerm(Representative(x)));
[ 1, -1, 1, 1, -1 ]
gap> List(ConjugacyClasses(a4),x->Size(x));
[ 1, 3, 4, 4 ]
gap> List(cls5,x->Size(x));
[ 1, 10, 15, 20, 20, 30, 24 ]
gap> List(cls5,x->SignPerm(Representative(x)));
[ 1, -1, 1, 1, -1, -1, 1 ]
gap> List(ConjugacyClasses(a5),x->Size(x));
[ 1, 15, 20, 12, 12 ]
```

共役類の計算 (5)

```
gap> List(ConjugacyClasses(c5),x->Size(x));
[ 1, 1, 1, 1, 1 ]
gap> List(ConjugacyClasses(d5),x->Size(x));
[ 1, 5, 2, 2 ]
gap> List(ConjugacyClasses(a5),x->Size(x));
[ 1, 15, 20, 12, 12 ]
gap> List(ConjugacyClasses(a7),x->Size(x));
[ 1, 105, 70, 210, 280, 630, 504, 360, 360 ]
gap> IsSimple(a7);
true
```

コメント 交代群 $A_n, (n \geq 5)$ は単純群であり，非自明な正規部分群を持たない．1 つの共役類 (クラス) に多くの元 (メンバー) が入っている事が，非可換性の強さを表わしている

シロー部分群の計算 (1)

S_5 の p-Sylow 部分群 $Sy(p)$ を求めてみる

```
gap> sy2:=SylowSubgroup(s5,2);
Group([ (1,2), (3,4), (1,3)(2,4) ])
gap> Elements(sy2);
[ (), (3,4), (1,2), (1,2)(3,4), (1,3)(2,4),
  (1,3,2,4), (1,4,2,3), (1,4)(2,3) ]
gap> List(Elements(sy4),x->Order(x));
[ 1, 2, 2, 2, 2, 4, 4, 2 ]
gap> sy2=d4;
false
gap> IsomorphismGroups(sy2,d4);
[ (1,2), (3,4), (1,3)(2,4) ] ->
  [ (1,2)(3,4), (1,4)(2,3), (1,3) ]
```

シロー部分群の計算 (2)

$Sy(2)$ と D_4 は S_5-共役 (i.e. $\exists g \in S_5$ s.t. $Sy(2) = gD_4g^{-1}$)

```
gap> IsConjugate(s5,sy2,d4);
true
gap> Size(ConjugateSubgroups(s5,sy2));
15
gap> sy3:=SylowSubgroup(s5,3);
Group([ (1,2,3) ])
gap> Size(ConjugateSubgroups(s5,sy3));
10
gap> sy5:=SylowSubgroup(s5,5);
Group([ (1,2,3,4,5) ])
gap> Size(ConjugateSubgroups(s5,sy5));
6
```

シローの定理から $Sy(3)$ (resp. $Sy(5)$) の共役部分群の個数は $1, 4, 10$ (resp. $1, 6$) 個に絞れるが，実際 10 個 (6 個)

可解群と単純群 (1)

5 次方程式	方程式のガロア群
$X^5 - X^3 - X^2 + X + 1 = 0$	$\longrightarrow S_5$
$X^5 + X^4 - 2X^2 - 2X - 2 = 0$	$\longrightarrow A_5$
$X^5 + X^4 + 2X^3 + 4X^2 + X + 1 = 0$	$\longrightarrow F_{20}$
$X^5 - X^3 - 2X^2 - 2X - 1 = 0$	$\longrightarrow D_5$
$X^5 + X^4 - 4X^3 - 3X^2 + 3X + 1 = 0$	$\longrightarrow C_5$

Fact. 四則演算と $\sqrt[n]{\ }$ を使った解の公式が存在する
$\Longleftrightarrow$ 方程式のガロア群が可解群 (=solvable group)

```
gap> f20:=Group((1,2,3,4,5),(1,2,4,3));
Group([ (1,2,3,4,5), (1,2,4,3) ])
gap> Order(f20);
20
gap> List([c5,d5,f20,a5,s5],x->IsSolvable(x));
[ true, true, true, false, false ]
```

可解群と単純群 (2)

$$F_{20} = \langle \sigma, \rho \rangle, \quad \sigma = (1\,2\,3\,4\,5), \quad \rho = (1\,2\,4\,3)$$
$$D_5 = \langle \sigma, \ \rho^2 \rangle, \rho^2 = \tau = (1\,4)(2\,3), \ C_5 = \langle \sigma \rangle$$

```
gap> CompositionSeries(f20);
[ Group([ (2,3,5,4), (2,5)(3,4), (1,2,3,4,5) ]),
  Group([ (2,5)(3,4), (1,2,3,4,5) ]),
  Group([ (1,2,3,4,5) ]), Group(()) ]
gap> DisplayCompositionSeries(f20);
G (3 gens, size 20)
 | Z(2)
S (2 gens, size 10)
 | Z(2)
S (1 gens, size 5)
 | Z(5)
1 (0 gens, size 1)
```

$$\{1\} \triangleleft C_5 \triangleleft D_5 \triangleleft F_{20}, \quad C_5/D_5 \cong C_2, D_5/F_{20} \cong C_2$$

可解群と単純群 (3)

S_4, A_4 もまた可解群である

```
gap> IsSolvable(s4);
true
gap> CompositionSeries(s4);
[ Group([ (3,4), (2,4,3), (1,4)(2,3), (1,3)(2,4) ]),
  Group([ (2,4,3), (1,4)(2,3), (1,3)(2,4) ]),
  Group([ (1,4)(2,3), (1,3)(2,4) ]),
  Group([ (1,3)(2,4) ]), Group(()) ]
gap> List(CompositionSeries(s4),x->Size(x));
[ 24, 12, 4, 2, 1 ]
gap> CompositionSeries(s4)[2]=a4;
true
gap> CompositionSeries(s4)[3]=v4;
true
```

$\{1\} \triangleleft C_2{}' \triangleleft V_4 \triangleleft A_4 \triangleleft S_4, \quad A_4/V_4 \cong C_3, S_4/A_4 \cong C_2$

可解群と単純群 (4)

S_n の可移部分群 (=transitive group) を求めてみる

```
gap> AllTransitiveGroups(NrMovedPoints,3);
[ A3, S3 ]
gap> List(last,x->Size(x));
[ 3, 6 ]
gap> AllTransitiveGroups(NrMovedPoints,4);
[ C(4) = 4, E(4) = 2[x]2, D(4), A4, S4 ]
gap> List(last,x->Size(x));
[ 4, 4, 8, 12, 24 ]
gap> AllTransitiveGroups(NrMovedPoints,5);
[ C(5) = 5, D(5) = 5:2, F(5) = 5:4, A5, S5 ]
gap> List(last,x->Size(x));
[ 5, 10, 20, 60, 120 ]
```

可解群と単純群 (5)

```
gap> all6:=AllTransitiveGroups(NrMovedPoints,6);
[ C(6) = 6 = 3[x]2, D_6(6) = [3]2, D(6) = S(3)[x]2,
  A_4(6) = [2^2]3, F_18(6) = [3^2]2 = 3 wr 2,
  2A_4(6) = [2^3]3 = 2 wr 3, S_4(6d) = [2^2]S(3),
  S_4(6c) = 1/2[2^3]S(3), F_18(6):2 = [1/2.S(3)^2]2,
  F_36(6) = 1/2[S(3)^2]2, 2S_4(6) = [2^3]S(3) = 2 wr S(3),
  L(6) = PSL(2,5) = A_5(6), F_36(6):2 = [S(3)^2]2
  = S(3) wr 2, L(6):2 = PGL(2,5) = S_5(6), A6, S6 ]
gap> List(last,x->Size(x));
[ 6, 6, 12, 12, 18, 24, 24, 24, 36, 36, 48, 60, 72,
  120, 360, 720 ]
gap> Filtered(all6,x->IsSimple(x)=true);
[ L(6) = PSL(2,5) = A_5(6), A6 ]
gap> List(last,x->Size(x));
[ 60, 360 ]
```

可解群と単純群 (6)

```
gap> all7:=AllTransitiveGroups(NrMovedPoints,7);
[ C(7) = 7, D(7) = 7:2, F_21(7) = 7:3,
  F_42(7) = 7:6, L(7) = L(3,2), A7, S7 ]
gap> List(last,x->Size(x));
[ 7, 14, 21, 42, 168, 2520, 5040 ]
gap> Filtered(all7,x->IsSimple(x)=true);
[ C(7) = 7, L(7) = L(3,2), A7 ]
gap> List(last,x->Size(x));
[ 7, 168, 2520 ]
gap> List([3..20],x->IsSimple(AlternatingGroup(x)));
[ false, false, true, true, true, true, true,
  true, true, true, true, true, true, true,
  true, true, true, true ]
```

Fact. (演習問題) A_n, $(n \geq 5)$ は (非可換) 単純群

位数の小さな有限群 (1)

命題. 固定した自然数 k に対し, 位数が k の有限群は (同型を同一視して) 有限個 (群表は有限種類しか作れない)

```
gap> AllSmallGroups(3);
[ <pc group of size 3 with 1 generators> ]
gap> AllSmallGroups(4);
[ <pc group of size 4 with 2 generators>,
  <pc group of size 4 with 2 generators> ]
gap> List([1..100],x->Size(AllSmallGroups(x)));
[ 1, 1, 1, 2, 1, 2, 1, 5, 2, 2, 1, 5, 1, 2, 1,
  14, 1, 5, 1, 5, 2, 2, 1, 15, 2, 2, 5, 4, 1,
  4, 1,51, 1, 2, 1, 14, 1, 2, 2, 14, 1, 6, 1, 4,
  2, 2, 1, 52, 2, 5, 1, 5, 1, 15, 2, 13, 2, 2, 1,
  13, 1, 2, 4, 267, 1, 4, 1, 5, 1, 4, 1, 50, 1,
  2, 3, 4, 1, 6, 1, 52, 15, 2, 1, 15, 1, 2, 1,
  12, 1, 10, 1, 4, 2, 2, 1, 231, 1, 5, 2, 16 ]
```

位数の小さな有限群 (2)

```
gap> all8:=AllSmallGroups(8);
[ <pc group of size 8 with 3 generators>,
  <pc group of size 8 with 3 generators>,
  <pc group of size 8 with 3 generators>,
  <pc group of size 8 with 3 generators>,
  <pc group of size 8 with 3 generators> ]
gap> List(all8,x->IsAbelian(x));
[ true, true, false, false, true ]
gap> List(all8,x->Exponent(x));
[ 8, 4, 4, 4, 2 ]
gap> IsomorphismGroups(all8[3],d4);
[ f1, f2, f3 ] -> [ (2,4), (1,2)(3,4), (1,3)(2,4) ]
gap> List(all8[4],x->Order(x));
[ 1, 2, 4, 4, 4, 4, 4, 4 ]
```

all8 は順に $\{C_8, C_2 \times C_4, D_4, Q_8, C_2 \times C_2 \times C_2\}$ と同型

位数の小さな有限群 (3)

```
gap> all16:=AllSmallGroups(16);;
gap> Length(all16);
14
gap> List(all16,x->IsAbelian(x));
[ true, true, false, false, true, false, false,
  false, false, true, false, false, false, true ]
gap> List(all16,x->IsSolvable(x));
[ true, true, true, true, true, true, true, true,
  true, true, true, true, true, true ]
gap> List(all16,x->Size(Center(x)));
[ 16, 16, 4, 4, 16, 4, 2, 2, 2, 16, 4, 4, 4, 16 ]
```

all16 の 14 個のうち，5 個がアーベル群．p-群は可解群だったので，残りの非可換群 9 個も可解群ではある

位数の小さな有限群 (4)

n	1	2	3	4	5	6
位数 n の群の同型類	$\{1\}$	C_2	C_3	C_4, V_4	C_5	C_6, S_3

7	8	9
C_7	$C_8, C_4 \times C_2, C_2 \times C_2 \times C_2, D_4, Q_8$	$C_9, C_3 \times C_3$

10	11	12	13	14
C_{10}, D_5	C_{11}	$C_{12}, C_2 \times C_6, D_6, A_4,$?	C_{13}	C_{14}, D_7

15	16
C_{15}	$C_{16}, C_2 \times C_8, (C_2)^2 \times C_4, C_4 \times C_4, (C_2)^4, D_8, D_4 \times C_2$

16	17	18
$Q_8 \times C_2$, ?, ?, ?, ?, ?, ?	C_{17}	$C_{18}, C_6 \times C_3, D_9, S_3 \times C_3,$?

19	20	21	22	23	$\cdots$
C_{19}	$C_{20}, C_2 \times C_{10}, D_{10}, F_{20},$?	C_{21}, F_{21}	C_{22}, D_{11}	C_{23}	?, $\ldots$

研究課題 位数の小さな群にはどの様なものがあるか？
? はどの様な群だろうか？
H_1, H_2 から大きい (非可換) 群 G を作るには? (半直積?, 群拡大?)